
4. BASIC BUILDING BLOCKS
This chapter introduces the concept of a procedure to assist in the solution of sub-problems, and shows how F's two types of procedures (functions and subroutines) and modules are used as the primary building blocks in well-designed programs.

4.1 Programs and modules

A sequence of statements starting with a program statement and ending with an end program statement is formally known as a main program or main program unit. 

There must be exactly one main program unit in an F program. However, F also has a second type of program unit, called a module. This is primarily a means of collecting together a set of related objects (procedures, variables, and data types).

The basic structure of module is similar to that of a main program unit. But the main difference is the one that a module does not contain any executable statement.


4.2 Procedures

A special section of programs, which is referred to whenever required, is known as procedures. Modules normally collect procedures together. 

Procedures fall into two broad categories, namely those which are written by the programmer and those which are part of the F itself. The latter group is known as intrinsic procedures, and their names are reserved words.

Procedures, which are not part of the F language, may be written in F or in other programming language known as external procedures. All procedures in an F program must be defined within modules.

There is also a further categorisation, based on their mode of use, into what are called subroutines and functions.

The purpose of a function is to take one or more values (called arguments) and create a single result. F contains a number of intrinsic

functions, which are a part of F language, for elementary mathematics, such as,

so on. There are 97 intrinsic functions available in F. Many of these functions can have arguments of more than one type. For example,

real :: x, y

y = abs(x)

will produce the absolute value of the real variable x.

4.3 Functions
The function statement, which is one of a sub-program of F, takes the form function 

name(d1, d2, ...) result(result_name)

where d1, d2, ... are dummy argument which are used when the function is executed. The result_name is the final result of the function after execution. The final statement of the function is the end function statement, taking the form,

end function name
As an example, a function, which calculates the cube root of a positive real numbers and a main program is given

module cube_root_calc_1

public :: cube_root

contains

function cube_root(x) result(root)

! Dummy argument declaration

real, intent(in)::x

! Result variable declaration

real::root

! Local variable declaration

real::log_x

! Calculate cube root by using logs

log_x = log(x)

root = exp(log_x/3.0)

end function cube_root

end module cube_root_calc_1

!******************************************************

program test_cube_root

use cube_root_calc_1

real :: x

print *," type real positive number"

read *, x

print *,"the cube root of x=",x," is",cube_root(x)

end program test_cube_root
There are three important points to notice about the function.

i. The variable log_x and root are not accessible from outside the function. They are called internal variables or local variables, and have no existence outside the function. Therefore the names log_x and root can be used in the main program or another procedure without any difficulty since they are isolated inside the function.

ii. Every function must have exactly one such result variable whose name must appear in the result clause of the function statement.

iii. The dummy argument, x, must be defined in the phrase intent(in) after the type declaration. This informs the compiler that the dummy argument can not be changed.

When the end function statement is obeyed it causes execution of the program to return to the point program at which the function was referenced. For example the statement

a = b*cube_root(c) + d

will cause the cube root of c to be calculated by the function , multiplied by b, added to the current value of d, then the sum is assigned to the variable a.

It is possible to write a function, which has no argument such as

function name() result name(result_name)

Although this kind of use is not very good, the function is referenced such as

a = b*name() + c.

As a result, writing its name, followed by any arguments it may have enclosed in parenthesis references a function. The execution of a function yields to a single value.

4.4 Subroutines

A subroutine is accessed by means of a call statement. This gives the name of the subroutine and list of arguments, which will be used to transmit information between the calling program unit and the subroutine:

call name(arg1, arg2, ....)

The call statement interrupts the execution of main program and the program flow goes to the statement contained within the subroutine name. When the subroutine has completed its task it returns to the place where it was called.
module various_roots_1

public :: roots

contains

subroutine roots(x,square_root,cube_root,fourth_root,fifth_root)

! subroutine to calculate various rots of a positive real   ! number supplied as the first argument, and returned them ! in the second to fifth arguments

!

! Dummy argument declaration

real, intent(in)::x
! Result variable declaration

real, intent(out)::square_root,cube_root,fourth_root,&

fifth_root

! Local variable declaration

real::log_x

! Calculate square root using intrinsic sqrt()

square_root = sqrt(x)

! Calculate other root by using logs

log_x = log(x)

cube_root = exp(log_x/3.0)

fourth_root = exp(log_x/4.0)

fifth_root = exp(log_x/5.0)

end subroutine roots

end module various_roots_1

!******************************************************

program various_roots

use various_roots_1

! A program to show the use of the subroutine roots

real :: x,root_2,root_3,root_4,root_5

print *," type real positive number"

read *, x

! obtain roots

call roots(x,root_2,root_3,root_4,root_5)

print *,"the square root of x=",x," is",root_2

print *,"the cube root of x=",x," is",root_3

print *,"the fourth root of x=",x," is",root_4

print *,"the fifth root of x=",x," is",root_5

end program various_roots
4.5 Actual and dummy arguments

When a function or subroutine is referenced, the information between the calling program unit and the subroutine or the function is passed through its arguments. The relationship between the actual arguments in the calling program unit and the dummy arguments in the subroutine or function is of vital importance. The order and types of the actual arguments must correspond exactly with the order and types of the corresponding dummy arguments.

intent(in),  intent(out),  intent(inout)  intent attribute is only used in the declaration of dummy arguments. 

An important F rule: In all cases use statements must immediately follow program, function or subroutine statement, and these must be followed in a procedure by the declaration of dummy arguments.

Example

Write a subroutine which will take two character arguments as input arguments, containing two names (a “first name” and a “family name”, respectively) and which will return a string containing the two names with exactly one space separating them.

Analysis & Solution

The major difficulty is in the declaration of the length of the two names in the subroutine. This difficulty can be handled by using an assumed-length character declaration in the F. his can only be used for declaring a dummy argument, and involves replacing the length specifier by an asterisk:

character (len=*) :: character_len_argument

This assumes that the length from the corresponding actual argument in the calling program is the same as the length of the dummy argument in the subroutine.
The other difficulty is the redundant spaces at the beginning or the end of the two names and then inserting exactly one character between them. The intrinsic function, trim, removes any trailing blanks from argument and adjustl moves argument enough spaces to the left to remove any leading blank.

module make_full_name_1

contains

subroutine get_full_name(first_name,last_name,full_name)

! Subroutine to join two names to form a full name with ! a single space between the first and last names

! Dummy argument declaration

character (len=*), intent(in) ::first_name,last_name

character (len=*), intent(out) ::full_name

! use adjustl to remove redundant leading blanks, and  ! trim to remove redundant blanks at the end of first  ! name

full_name = trim(adjustl(first_name)) // " " //&

adjustl(last_name)

end subroutine get_full_name

end module make_full_name_1

! ***************************************************

program make_full_name

use make_full_name_1

! Variable declaration
character (len=80)::name_1,name_2,name

! read the first name

print *,"enter the first name"

read *,name_1

! read the surname
print *,"enter the surname"

read *,name_2

! combine two names by using the subroutine named 

! get_fuul_name

call get_full_name(name_1,name_2,name)

print *,"the combination of the two names is => ",name

end program make_full_name
Although the dummy arguments are declared to be of assumed length, the corresponding actual arguments are declared with specific lengths in the calling program (here main program) unit.

The result of the adjustl function has been used as the argument to trim. The first function, adjustl, therefore moves its argument, first_name, to the left to eliminate any leading blanks, and the second trim takes the result and removes any trailing blanks.

4.6 Local and global objects

Each program unit is only aware of any variables declared within the program unit itself, its local variables, and dummy arguments. No clash of names.

Public statement is used to specify the names of procedures to be made available to any program unit using the module. Any public statements must appear before any declaration statements.

A module may also be used to make variables and constants available to several procedures. This can be performed using a public attribute in the declaration statement. 

For example if one wished to use the values of (, g (acceleration of gravity) and e in a number of different procedures, a simple module performing this would be,
module Natural_Constants

real,parameter,public::pi=3.1415927,g=9.81,e=2.7182818

end module Natural_Constants
In order to obtain access to the constants defined in the module, Natural_Constants, the use statement must appear immediately after the initial statement, which may be a module, a procedure or a main program as shown in the module_example_1 example.
program module_example_1

Use Natural_Constants

print *,g,pi

end program module_example_1
In this example, the values, g and (, are printed out without any assignment to them but the module, Natural_Constants, must be located before program module_example_1.

Host association:

Any variables or constants declared in a module before the contains statement is automatically available to the procedures within that module. However, if we only want to make the variables or constants available only within the module but not outside it when the model is used elsewhere, we must use private attribute in the declarations.

module Exmpl
public :: sub1,sub2

real,parameter,public::pi=3.1415927,e=2.7182818

real, private :: glb_var1, glb_var2 !Only within the module
contains

subroutine sub1(a,b,c)

.

.

c=pi*a*b

.

.

end subroutine  sub2

subroutine sub1(x,y,z)

.

.

glob_var1=x*. glob_var2

.

end subroutine  sub2

end module Exmpl
Modules can use other modules but not themselves.

The private statement must be used in a module if it accesses to another module via use statement in order not to make available the items of the used module to other program units when the model itself is used by other program units.

module another_mdl

use Natural_Constants

private

real, parameter, public :: pi_by_two=0.5*pi

.

.

end module another_mdl

Modules can also be collected in a module. It must contain a sequence of use statements followed by a public statement. In this case everything accessible from the individual modules is to be acessible from this module when it is used.

module super_mdl

use mdl_1

use mdl_2

.

.

.

public

end module super_mdl

4.7 Giving procedure variables an initial value

If there is a situation when a variable in a procedure may be required to have an initial value on the first reference to the procedure the save attribute can be used.

real, save :: b=1.23

integer, save :: count=0
In this program part example, the initial values are assigned to the variable b as real and the variable count as integer.

4.8 Procedures as an aid to program structure

The modular program development, which is the key concept of the software engineering, enables us to break the design of a program into several smaller, more manageable sections known as modules, and procedures. But this requires much more experience and effort on programming.

